

Physics

HP COMPUTER CURRICULUM
Mechanics

STUDENT LAB BOOK

HEWLETT  PACKARD

Hewlett-Packard
Computer Curriculum Series

physics
STUDENT LAB BOOK

mechanics

by Herbert D. Peckham
Gavilan College

edited by Briana Burns and Jean Danver
Hewlett-Packard

Hewlett-Packard Company
11000 Wolfe Road
Cupertino, California 95014

© Copyright, 1972, by
HEWLETT-PACKARD COMPANY
Cupertino, California
Printed in the U.S.A.

The Hewlett-Packard Computer Curriculum Series represents the published results of a Curriculum Development Project sponsored by the Data Systems Division of Hewlett-Packard Company. This project is under the directorship of Jean H. Danver.

This material is designed to be used with any Hewlett-Packard system with the BASIC programming language such as the 9830A, Educational BASIC, and the 2000 and 3000 series systems.

The Hewlett-Packard Company makes no warranty expressed or implied, and assumes no responsibility in connection with the operation of the program material contained herein.

Copyright © 1972 by Hewlett-Packard Company, Cupertino, California. All rights reserved. No part of this publication may be reproduced, stored in a retrieval system (e.g., in memory, disc or core) or be transmitted by any means, electronic, mechanical, photocopy, recording or otherwise, without prior written permission from the publisher.

Printed in the U.S.A.

TABLE OF CONTENTS

- 1 INTRODUCTION
- 3 MECHANICS
- 5 RATES
 - 7 Exercise 1 — Computing Velocity
 - 7 Exercise 2 — Applications
 - 9 Exercise 3 — Automatic Accuracy
 - Advanced Exercises*
 - 9 Exercise 4 — Plotting Results
 - 9 Exercise 5 — Rates and Trigonometric Functions
- 11 FROM RATES TO DISPLACEMENTS
 - 13 Exercise 6 — Displacement
 - 14 Exercise 7 — Increasing the Accuracy
 - 14 Exercise 8 — Finding Displacement from Acceleration
 - Advanced Exercises*
 - 14 Exercise 9 — Average Values of Velocity & Acceleration
 - 14 Exercise 10 — A Generalization
- 15 NEWTON'S SECOND LAW
 - 16 Exercise 11 — Force and Motion
 - 18 Exercise 12 — Converging on Accuracy
 - 18 Exercise 13 — A Variable Force
 - Advanced Exercises*
 - 18 Exercise 14 — A Challenge
 - 18 Exercise 15 — Barrier Penetration
- 19 HALF STEP METHOD
 - 19 Exercise 16 — Half Step Computation
 - 19 Exercise 17 — Half Step Computation
- 21 THE HARMONIC OSCILLATOR
 - 21 Exercise 18 — Harmonic Oscillator
 - 21 Exercise 19 — Half Step Computation
 - 21 Exercise 20 — Changing the Mass
 - Advanced Exercise*
 - 22 Exercise 21 — The Pendulum

TABLE OF CONTENTS (Continued)

- 23 MORE COMPLICATED FORCES
- 23 Exercise 22 – Damped Harmonic Oscillator
- 23 Exercise 23 – An Experiment
 - Advanced Exercise*
- 24 Exercise 24 – Driven, Damped Harmonic Oscillator

- 25 ORBITAL MOTION
- 27 Exercise 25 – Orbital Motion
- 27 Exercise 26 – Experimentation
- 27 Exercise 27 – A Challenge
 - Advanced Exercise*
- 27 Exercise 28 – A Double Force Center

- 29 APPLICATION PROGRAM

INTRODUCTION

This Physics Lab Book was developed to provide you the opportunity to use a computer as a problem solving tool. You will write computer programs which will enable you to investigate the great ideas of physics. Using just one program, you will be able to perform many different experiments. You can make your own generalizations.

To use the Lab Book for Mechanics, you will need the following. First, you should have a background in algebra and some trigonometry. Second, the Lab Book assumes that you already know how to write a simple computer program using the BASIC language. If you do not, you will want to study this before you attempt the material. Consult the BASIC Manual for the computer you use. Last, use of this Lab Book requires that you have access to a computer for at least two hours per week. If more time is available, you may be able to experiment further on your own, either to improve your program or to investigate other aspects of physics that interest you.

As you will discover, there is no one "right" way to use a computer as a problem solving tool. There are many different ways to solve one problem by programming. Experiment and learn as you go. You'll find you are learning something new each time, both about your subject matter and about using the computer to solve problems.

This book was designed to help you by providing several different kinds of material. First, there are the exercises with the preparatory explanatory material. These exercises are sequenced so that you can apply what you have learned in the previous problem in solving the next one. Often you can take your preceding program and simply add to it to create a program that will provide answers to the more general or more advanced problem.

Sections of advanced problems are provided for any student interested in further work in this area. You may wish to tackle these after you have completed the basic exercises.

An example program and flow chart follow the exercises. You may wish to review this flow chart and program before you begin using the Lab Book. The example demonstrates planning a solution (the flow chart) and the coding of the solution (the program). When you do begin using the Lab Book, you may choose to flow chart your solutions first. This is good programming practice. Drawing the flow chart provides a check of your logic, and the finished flow chart can then be used as a guide in writing each step of your program.

NOTES

MECHANICS

If physics is the *king* of scientific knowledge, then mechanics must certainly be the *queen* of physics. No other part of physics has such a long and rich history. The father of mechanics as we know it today was Galileo. His three laws of motion were first presented in 1686 in his *Principia Mathematica Philosophiae Naturalis* and laid the foundation for mechanics. In the year of Galileo's death, the person responsible for the full flowering of mechanics was born. This was Isaac Newton — one of the most famous scientists of all time. Between the ages of 23 and 25 he made an incredible number of most important advances in physics. Among these was the invention of calculus which he needed to explain the motion of the planets, his universal law of gravitation, and his three laws of motion. A steady succession of mathematicians and physicists elaborated upon and extended the structure of Newton's mechanics throughout the 18th and 19th centuries. Today the subject of classical mechanics is one of the most complete and well investigated parts of physics. It is not accidental that most studies of physics begin with mechanics.

NOTES

RATES

The concept of rate is one which finds its way into every part of physics. It is particularly important in the study of mechanics. Consequently, we should look carefully at the concept and be certain that it is thoroughly understood.

Suppose you watch the second hand on a watch move around the dial. How fast is it moving? Or, consider the motion of an automobile as it slows down on the highway. How fast is it stopping? These examples all involve the central idea of *rate*.

By rate we mean *change in a variable with respect to a change in another variable*. For example, we could look at the rate of change of displacement with respect to time which is defined as velocity. Or, similarly, acceleration is the rate of change of velocity with respect to time. Electrical current is the rate of change of charge with respect to time, and so on. To make the concept clear, we will look at an example in some detail.

Consider an object moving along a straight line such that its position is given by

$$x = t^3 + 2. \quad (1)$$

It is easy to determine *where* the object is at any time t since we have only to substitute the appropriate value of t into (1). The more interesting question is *how fast* is the object moving at some value of t ? Or, phrased differently, what is the *rate of change of x with respect to t* ? The computer provides an effective tool to investigate this and other similar questions.

One way to attack the problem is to locate the object at two different times and then compute the average rate of change as the difference in position divided by the corresponding difference in time. In this example we would get the average velocity over the time interval.

This concept of average velocity is in agreement with our common experience with automobiles. If, for example, we are at point A which is 10 miles from the starting point 1 hour after beginning a journey along a straight line path and are at point B which is 30 miles from the starting point 2 hours after starting, we should all agree that the average velocity between A and B is 20 miles per hour. We know nothing about the instantaneous velocity at any specific time in the 2-hour interval. All we can say is that an average of 20 miles was covered in one hour.

Returning to the object described by (1), if we want to find the average velocity of the object over the time span from $t = 1$ to $t = 3$, we would obtain

$$v_{ave} = \frac{(3^3 + 2) - (1^3 + 2)}{3 - 1} = 13 \quad . \quad (2)$$

Now, suppose we repeat the process except over the time span from $t = 1.5$ to $t = 2.5$. The result is

$$v_{ave} = \frac{(2.5^3 + 2) - (1.5^3 + 2)}{2.5 - 1.5} = 12.25 \quad . \quad (3)$$

If we continue the process, computing the average velocity for time intervals which narrow down around $t = 2$, it seems reasonable to conclude that we are getting progressively closer to the notion of *how fast* the object is moving at $t = 2$. Thus, the critical difference between instantaneous and average velocities is the size of the time increment used in the computation. If the time increment is sufficiently small, there is no sensible difference between the two. The *instantaneous velocity* tells us how fast the object is moving at a particular instant in time. On the other hand, the *average velocity* tells us how fast the object is moving, averaged over some *time interval*.

We can put this in a more convenient and powerful mathematical notation. Suppose that $x = f(t)$. This is read as "x equals a function of t." The $f(t)$ denotes any function of time. It is not necessary at this point to define which particular one is under consideration. A function must be defined, of course, in the computer program. We want to compute the average velocity over a time span Δt divided equally around t . (Δt is read "delta t" and stands for an increment, or small value of time.) If we refer to the calculations in (2) or (3) we can easily see that the correct relationship which we are seeking is

$$v_{ave} = \frac{f(t + \Delta t/2) - f(t - \Delta t/2)}{\Delta t} \quad (4)$$

Figure 1 shows a BASIC program to compute v_{ave} assuming that $f(t) = t^3 + 2$. With this program we can investigate the velocity of the object.

```

100  REM PROGRAM FOR AVERAGE VELOCITY
110  PRINT "INPUT VALUE OF T DESIRED";
120  INPUT T
130  PRINT "INPUT VALUE OF D";
140  INPUT D
150  DEF FNA(T)=T^3+2
160  LET V=(FNA(T+D/2)-FNA(T-D/2))/D
170  PRINT "AVERAGE VELOCITY IS ";V
180  PRINT
190  GOTO 110
999  END

```

Figure 1. Program to Compute Velocities

The printout in Figure 2 shows the results for several values of t . In each case, decreasing values of D (which stands for Δt) are used until there is no further change in v_{ave} . At this point we are certain that we do have the instantaneous velocity at the point in time under consideration. Note that if we keep decreasing D , v_{ave} begins to change. This is caused by round-off error in the particular computer used here.

Exercise 1 — Computing Velocity

Use the program in Figure 1 to determine the instantaneous velocities of an object moving according to $x = \sin(t) + t^2$ at $t = 1$, and $t = 2$.

Exercise 2 — Applications

Use the program in Figure 1 to investigate the following functions:

$$(a) \quad x = 3t^3 - 4t^2 + 5$$

$$(b) \quad x = e^t + t$$

$$(c) \quad x = \cos(t^3) - e^{\sin(t)}$$

RUN

INPUT VALUE OF T DESIRED?1

INPUT VALUE OF D?1

AVERAGE VELOCITY IS 3.25

INPUT VALUE OF T DESIRED?1

INPUT VALUE OF D?.1

AVERAGE VELOCITY IS 3.0025

INPUT VALUE OF T DESIRED?1

INPUT VALUE OF D?.01

AVERAGE VELOCITY IS 3.00007

INPUT VALUE OF T DESIRED?1

INPUT VALUE OF D?.001

AVERAGE VELOCITY IS 2.99978

INPUT VALUE OF T DESIRED?1

INPUT VALUE OF D?.0001

AVERAGE VELOCITY IS 2.99931

INPUT VALUE OF T DESIRED?1

INPUT VALUE OF D?.00001

AVERAGE VELOCITY IS 3.00407

INPUT VALUE OF T DESIRED?2

INPUT VALUE OF D?1

AVERAGE VELOCITY IS 12.25

INPUT VALUE OF T DESIRED?2

INPUT VALUE OF D?.1

AVERAGE VELOCITY IS 12.0025

INPUT VALUE OF T DESIRED?2

INPUT VALUE OF D?.01

AVERAGE VELOCITY IS 12.0001

INPUT VALUE OF T DESIRED?2

INPUT VALUE OF D?.001

AVERAGE VELOCITY IS 12.001

INPUT VALUE OF T DESIRED?2

INPUT VALUE OF D?.0001

AVERAGE VELOCITY IS 12.0163

Figure 2. Printout from Velocity Program

Exercise 3 — Automatic Accuracy

Devise a way to modify the program in Figure 1 such that N values of t are read from DATA statements, and the instantaneous velocity, correct to 4 significant digits, is computed and printed out for each t . Assume that the appropriate $f(t)$ is defined in the DEF statement in the program. Test your revised program on $x = t^2$; $t = 1, 2, 3, 4$.

Exercise 4 — Advanced — Plotting Results

Run the program from Exercise 3, except use $x = t$. Plot the resultant values of instantaneous velocity (as well as those from Exercise 3) versus time. See if you can detect any relationship that may be present. Do not hesitate to ask your instructor for assistance.

Exercise 5 — Advanced — Rates and Trigonometric Functions

Run the program from Exercise 3 with $x = \sin(t)$; $t = 0, .1, .2, \dots, 1.0$. Plot t , $\cos(t)$, $\sin(t)$, and the values of velocity on the same graph. Can you write an expression for v as a function of time?

NOTES

FROM RATES TO DISPLACEMENTS

The preceding discussion can be turned around. Suppose that we know the instantaneous velocity of an object is given by

$$v = 2t^2 + 4 \quad (5)$$

and that when $t = 0$, $x = 0$. It is easy to determine *how fast* the object is moving since we have only to substitute the appropriate value of t into (5). The interesting question now is, *where* is the object at time t ?

Remember that in the previous discussion we were able to develop the idea of average velocity as a change in distance divided by the associated change in time, or

$$v_{ave} = \Delta x / \Delta t \quad (6)$$

where Δx stands for change in distance, and Δt stands for change in time. We know also that if Δt is small enough there is no sensible difference between the average and instantaneous values of the velocity in the time interval Δt . Consequently, for small enough Δt , we can write

$$v = \Delta x / \Delta t \quad (7)$$

where v represents instantaneous velocity. Substituting (5) in this we have

$$\Delta x = (2t^2 + 4)\Delta t \quad (8)$$

But this gives only the *change* in position that took place during the time interval Δt . A little thought reveals that the correct expression for the new position is

$$x_{\text{new}} = x_{\text{old}} + \Delta x \quad (9)$$

or

$$x_{\text{new}} = x_{\text{old}} + (2t^2 + 4)\Delta t. \quad (10)$$

If, when $t = 0$, $x_{\text{old}} = 0$, and if $t = 0.1$, we have

$$x_{\text{new}} = 0 + (2(0)^2 + 4)(.1) = 0.4 \quad (11)$$

If we repeat the process once more we have

$$x_{\text{new}} = 0.4 + (2(.1)^2 + 4)(.1) = 0.802 \quad (12)$$

We could go on, stepping forward in time, computing the new position at each step, but we must pause to consider carefully what is taking place. We compute the change in position taking place during a time interval Δt . However, we use the value of the velocity at the beginning of the time interval *and assume that it remains constant*. Obviously this is in general not true. However, if Δt is kept small the error can be controlled.

```

100  REM PROGRAM TO FIND DISPLACEMENT
110  REM GIVEN VELOCITY
120  PRINT "INPUT INITIAL VALUE OF X  ";
130  INPUT X0
140  PRINT "INPUT TIME INCREMENT  ";
150  INPUT D
160  DEF FNA(T)=2*T^2+4
170  FOR T=0 TO 1 STEP D
180  PRINT T,X0
190  LET X1=X0+FNA(T)*D
200  LET X0=X1
210  NEXT T
999  END

```

Figure 3. Program to Compute Displacements

The program in Figure 3 carries out this process in time steps of D from 0 to 1. X_0 stands for the old displacement, while X_1 stands for the new displacement. As before, D is the symbol for Δt . Figure 4 gives the printouts for $D = 0.2, 0.1$, and 0.05 . Note that the values of the displacements change as D is changed. Generally, the smaller D is, the more accurate the displacement is *down to some critical point*. If D is decreased past this point, the error starts to build up due to round off error in the computer.

RUN

INPUT INITIAL VALUE OF X ?0
INPUT TIME INCREMENT ? .05

0	0
.05	.2
.1	.40025
.15	.60125
.2	.8035
.25	1.0075
.3	1.21375
.35	1.42275
.4	1.635
.45	1.851
.5	2.07125
.55	2.29625
.6	2.5265
.65	2.7625
.7	3.00475
.75	3.25375
.8	3.51
.85	3.774
.9	4.04625
.95	4.32725
1.	4.6175

DONE

RUN

INPUT INITIAL VALUE OF X ?0
INPUT TIME INCREMENT ? .2

0	0
.2	.8
.4	1.616
.6	2.48
.8	3.424
1	4.48

DONE

RUN

INPUT INITIAL VALUE OF X ?0
INPUT TIME INCREMENT ? .1

0	0
.1	.4
.2	.802
.3	1.21
.4	1.628
.5	2.06
.6	2.51
.7	2.982
.8	3.48
.9	4.008
1	4.57

DONE

Figure 4. Printouts from Displacement Program

Exercise 6 — Displacement

Let $v = \cos(t)$, and $x = 0$ when $t = 0$. Use the program in Figure 3 to find how x varies with time.

Exercise 7 — Increasing the Accuracy

If we know the expression for v , as in Exercise 6, we can compute the velocity at the end of each time increment as well as at the beginning. If the average of these two velocities is used to compute the change in x taking place during that increment, the error should decrease significantly. Write a program to do this. Run the problem in Exercise 6 again using this new program.

Exercise 8 — Finding Displacement From Acceleration

We know that $v = \Delta x / \Delta t$, but acceleration in turn is defined by $a = \Delta v / \Delta t$. Thus, we can begin with a known acceleration, compute Δv from $\Delta v = a \Delta t$, and compute Δx from $\Delta x = v \Delta t$. All we need to know is $a = f(t)$, and the values of v and x at $t = 0$. Write a BASIC program to carry out this process. Test your program on $a = t$, $\Delta t = 0.1$, for t from 0 to 1, with $x = 0$ and $v = 0$ at $t = 0$. Have the program compute and print out t , v , and x at each step.

Exercise 9 — Advanced — Average Values of Velocity and Acceleration

Arrange the program in Exercise 8 to use the average value of a and v in each time step as in Exercise 7.

Exercise 10 — Advanced — A Generalization

Run the program developed for Exercise 9 in time steps of 0.1 from $t = 0$ to $t = 4$. Plot a , v , and x for each value of t . Can you generalize your results?

NEWTON'S SECOND LAW

Now that we have discussed rates we can profitably proceed to Newton's second law. It is very difficult to think of a single man whose work has had a more direct and far reaching effect upon western civilization than Isaac Newton's. His laws occupy a position of pivotal importance in mechanics, which is probably the most well developed part of physics, which in turn lies at the heart of much of today's technology.

Newton's Second Law is

$$F = ma \quad (13)$$

Unfortunately, it is possible to get the impression that this is nothing more than an algebraic equality involving three variables stating that force equals mass times acceleration. To do so is to miss completely the rich content in the law. With the foundation laid in the last section we can proceed to discover much more than a simple equality.

Recall that we can write acceleration as $a = \Delta v / \Delta t$, and velocity as $v = \Delta x / \Delta t$. The only restriction we must be careful to observe is that Δt must be small enough so that there is no significant difference between average and instantaneous values of the acceleration and velocity. Replacing a in (13) with $\Delta v / \Delta t$ we can write

$$\frac{\Delta v}{\Delta t} = \frac{1}{m} F \quad (14)$$

and

$$\frac{\Delta x}{\Delta t} = v \quad (15)$$

Equation (14) gives Newton's Second Law in a form not generally seen in introductory books. The important difference between (14) and (13) is the idea of *rate* which is apparent in (14). Velocity is written to emphasize the same idea in (15). To sense the significance of this we must ask what is the point in the Second Law? What kind of answers does it provide?

The fundamental problem is to predict the motion of an object if we know the forces acting on it as well as the object's position and velocity at some particular instant. In simpler terms, we are trying to find out how things move when forces act on them. Newton's Second Law provides the key.

If (14) and (15) are rearranged slightly we can compute the change in velocity and position during a time increment. If the changes are added to the previous values of velocity and position we can compute new values whereupon the process is repeated, stepping forward in time. The equations needed to carry out this process are

$$v_{\text{new}} = v_{\text{old}} + \frac{1}{m} F \Delta t \quad (16)$$

and

$$x_{\text{new}} = x_{\text{old}} + v_{\text{old}} \Delta t \quad (17)$$

These simple equations, coupled with the calculating speed of the digital computer, are much more powerful than you might imagine. All we need is the initial position and velocity and the force acting during each time increment. With this information we can indeed find out *how things move*, and can solve problems which are hopeless with analytical methods.

At this point, an example should firm up the ideas. Suppose that we want to find out how an object of mass $m = 1$, located initially at $x = 2$, with velocity $v = -4$ will move if subjected to a constant force of $F = +2$. Assume that the quantities above are given in some consistent system of units. You may wish to discuss the question of units further with your instructor. We shall not be overly concerned with units in this treatment.

The program in Figure 5 solves equations (16) and (17) and follows the motion of the object. The program uses D for the time step, $X0$ and $V0$ for the old values of position and velocity, and $X1$ and $V1$ for the new position and velocity respectively. The program is arranged to print out the time, velocity, and position every N computations. The time limit on the computation is L . The printout for the program is contained in Figure 6. Study both the program and printout until you are certain you understand the process.

Exercise 11 – Force and Motion

Run the program in Figure 5 with $m = 2$. Run it again with m returned to the initial value, but with $F = 8$. Plot the velocity and position data from Figure 6 as well as the data from this exercise versus time. Explain your results.

```

100  REM NEWTON'S SECOND LAW
110  READ X0,V0,M
120  READ D,N,L
130  LET F=2
140  PRINT
150  PRINT "T","V","X"
160  PRINT
170  LET C=N
180  FOR T=0 TO L STEP D
190  IF C<N THEN 220
200  PRINT T,V0,X0
210  LET C=0
220  LET V1=V0+F*D/M
230  LET X1=X0+V0*D
240  LET V0=V1
250  LET X0=X1
260  LET C=C+1
270  NEXT T
800  DATA 2,-4,1
801  DATA .1,10,5
999  END

```

Figure 5. Program for Newton's Second Law

RUN

T	V	X
0	-4	2
1	-2.	-1.1
2.	-1.84774E-06	-2.2
3.	2.	-1.3
4.	4.	1.59999

DONE

Figure 6. Printout from Newton's Second Law Program

Exercise 12 — Converging on Accuracy

Run the program in Figure with the following values of D and N :

- (a) $D = .1, N = 10$
- (b) $D = .05, N = 20$
- (c) $D = .01, N = 100$
- (d) $D = .005, N = 200$

Compare the four printouts and see if you can explain what is taking place. Which set of data do you feel is the most accurate?

Exercise 13 — A Variable Force

Run the program in Figure 5 except set $F = 4\text{Cos}(t)$. Note that F must now be computed inside the time loop. Plot your results.

Exercise 14 — Advanced — A Challenge

Suppose that for $|x| < 5$, $F = 0$; for $x >= 5$, $F = -4$; for $x <= -5$, $F = +4$; $m = 1$, $x_0 = 0$ and $v_0 = +5$. Develop a program to follow the object. Plot your results.

Exercise 15 — Advanced — Barrier Penetration

Given a force described by $F = 0$ for $x < 0$ or $x > 5$, and $F = -5$ for $0 \leq x \leq 5$. If the initial position is 0 and $m = 1$, find the initial velocity such that the object arrives at $x = +5$ with a positive velocity less than .1. Do the problem by a trial and error technique. This is a simulation of a barrier penetration.

HALF STEP METHOD

It may have occurred to you that we have been using a rather simple-minded method to solve the exercises up to this point. In all instances we have computed the change in velocity, for example, assuming that the velocity remained constant throughout the time interval. Except in trivial cases, this isn't true. We have been able to keep the error down by using very small time increments. However, it turns out that a very simple device may be used to improve the method. This device is known as the *half step method*.

The strategy is to compute a single change in velocity at the beginning of the program utilizing a half time step. If subsequently we proceed as usual, the velocity and position will be out of step by one half time step. The advantage lies in the fact that now the change in position is computed with a value of velocity that is half way through the time step. This simple procedure produces a dramatic increase in accuracy.

Exercise 16 – Half Step Computation

Modify the program in Figure 5 to include an initial half step computation in the velocity. Compare your results to those obtained initially.

Exercise 17 – Half Step Computation

Using the program developed in Exercise 16, rework Exercise 12 and compare your results to those obtained initially.

NOTES

THE HARMONIC OSCILLATOR

A number of interesting phenomena in physics are associated with the concept of the *harmonic oscillator*. The simplest type of harmonic oscillator is a mass hanging on a spring. We assume that the force generated when the spring is stretched is governed by Hooke's law which states that the force is proportional to the displacement of the spring from the equilibrium position and is directed opposite to the displacement.

$$F = -kx \quad (18)$$

Here, k is the constant of proportionality, and x measures the displacement of the system from equilibrium. But we already have a computer program which can be used to solve for the motion of the mass on the spring since we are given the force in (18) and we know how to solve Newton's Second Law.

Exercise 18 — Harmonic Oscillator

Use the program in Figure 5 to solve the harmonic oscillator problem if $m = k = x_0 = 1$, and $v_0 = 0$. Note that the force depends upon x and consequently must be inside the calculation loop. Use a time step of $\pi/100$, print out every ten steps, and follow the motion of a time limit of 2π . The easiest way to handle π in your program is to set $P = 3.14159$ in the program and then work with multiples of this value.

Exercise 19 — Half Step Computation

Modify the program in Exercise 18 to include an initial half step in the velocity calculation. Run the program and plot the results. Write down a mathematical function which would generate the response you have plotted. From this can you make any generalizations about the behavior of harmonic oscillators?

Exercise 20 — Changing the Mass

Run the program from Exercise 19 with $m = 1/2$, and $m = 2$. Do the results seem consistent with your understanding of rates and Newton's Second Law?

Exercise 21 – Advanced – The Pendulum

Consult your physics textbook about the motion of a simple pendulum. Write a computer program to follow the motion once the initial values of angle and velocity have been specified.

MORE COMPLICATED FORCES

Once we have a good program for the harmonic oscillator, it is very easy to modify the program to account for more interesting cases. For example, if you observe a mass oscillating on a spring it is clear that the oscillations gradually die out. Yet, there is nothing in the computer results so far to account for this. The problem is that forces are present which are not included in the program. The damping force is generally described by

$$F = -\beta v \quad (19)$$

It is extremely important to note that if, for example, the damping had been proportional to the square of the velocity, it would have made little difference in the manner in which we accounted for damping in the computer program. This is emphatically not so for analytical methods. Minor changes in the description of the forces which govern a problem may well produce equations which cannot be solved by analytical methods. However, with the computer we are in the fortunate situation of requiring only a description of the force which we can compute at the proper place in the program, and we can then usually reach a solution with no difficulty.

Exercise 22 — Damped Harmonic Oscillator

To examine damping, use $F = -kx - \beta v$ in the program in Exercise 19. Let $k = m = \beta = 1$. Run the program and plot the results.

Exercise 23 — An Experiment

Set up an experimental mass spring system and measure m and k . Set the system into oscillation and measure the time required for the amplitude of the oscillations to decrease to half their initial value. Using a trial and error approach, input values of β into the program from Exercise 22 until the predicted time for the amplitude to decrease to half its initial value agrees with the experimental measurement. You will be using the computer to discover the damping constant of proportionality. Since your program will probably have to follow several cycles, use the half step method with a small time step.

Exercise 24 – Advanced – Driven, Damped Harmonic Oscillator

*Use the same conditions as in Exercise 22, except add the force $F = 2\cos(t)$.
Compare your results to those of Exercise 22.*

ORBITAL MOTION

One of the greatest achievements of Newtonian mechanics was the discovery of the planet Neptune in 1846 on paper before a visual sighting through a telescope was made. The new planet was *invented* to account for the observed perturbations in the orbit of the planet Uranus. When telescopes were pointed to the predicted location, the new planet was indeed there.

We can easily adapt the developments thus far to follow the motion of planets or other bodies moving under the influence of gravitational forces. The main difference is that now we can have motion in two dimensions. Newton's law of gravitation states that

$$F = \frac{GMm}{r^2} \quad (20)$$

G is the universal gravitation constant, M is the mass of one body, m the mass of another, and r is the distance between them. Moreover the force is directed from one body towards the other along a line joining them.

Figure 7 portrays an object of mass M which we will assume is fixed at the origin, and a second object of mass m located at the point (x, y) which is assumed to be




Figure 7. The Gravitational Forces

moving under the influence of the gravitational force between the two objects. The strategy we will use is to split the motion up into that parallel to the x axis and that parallel to the y axis. Thus we must resolve the gravitational force F into its x and y components. If we do this we have

$$F_x = -\frac{GMm}{r^2} \cos(\theta) = -\frac{GMm}{r^2} \left(\frac{x}{r}\right) = -\frac{GMmx}{r^3} \quad (21)$$

$$F_y = -\frac{GMm}{r^2} \sin(\theta) = -\frac{GMm}{r^2} \left(\frac{y}{r}\right) = -\frac{GMmy}{r^3} \quad (22)$$

$$r = \sqrt{x^2 + y^2} \quad (23)$$

The two forces represented by (21) and (22) are *coupled*. In other words, a change in y can produce a change in F_x and a change in x can produce a change in F_y . The coupling occurs because of the dependence of r in (23) on both x and y.

If we now use these forces, we can write a set of equations equivalent to (16) and (18) for both the x and y directions.

$$v_{x_{\text{new}}} = v_{x_{\text{old}}} + (F_x/m)\Delta t \quad (24)$$

$$x_{\text{new}} = x_{\text{old}} + (v_{x_{\text{old}}})\Delta t \quad (25)$$

$$v_{y_{\text{new}}} = v_{y_{\text{old}}} + (F_y/m)\Delta t \quad (26)$$

$$y_{\text{new}} = y_{\text{old}} + (v_{y_{\text{old}}})\Delta t \quad (27)$$

Note that m in (24) and (26) has no effect since it cancels out when F_x and F_y are used according to (21) and (22). To simplify matters and to avoid large numbers we will assume that we are in a coordinate system such that $GM = 1$. This enables us to study the shapes of various orbits without becoming involved in messy mathematics. Now we require initial position and velocity in both the x and y directions. The procedure is precisely the same as for the exercises involving Newton's Second Law, except we have two sets of equations.

Exercise 25 – Orbital Motion

Write a program to follow the motion of an object moving under the influence of gravity. Assume that $GM = 1$. Use an initial half step computation for both x and y velocities. To test your program, assume $x_0 = 1$, $y_0 = 0$, $v_{x_0} = 0$, and $v_{y_0} = 1$. These conditions produce a circular orbit. Consequently, with the initial half step, and a reasonably small time step, your program should produce a circular orbit. You can check on this by having the program output r at each step.

Exercise 26 – Experimentation

Using the program from Exercise 25, try out various initial positions and velocities to acquire a feel for the types of motion that can be produced.

Exercise 27 – A Challenge

Use the program from Exercise 25 with $x_0 = 1$, $y_0 = 0$ and both initial velocities equal to zero. Run the program and explain the results.

Exercise 28 – Advanced – A Double Force Center

Repeat the development for gravitational motion but have two fixed objects of mass M as well as the moving object of mass m . Develop a computer program to handle this and investigate the types of motion that result. You must fix the two masses with mass M at two points in space. Compare the orbits you get with those from Exercise 26.

NOTES


APPLICATION PROGRAM

The program for the solution of Newton's Second Law which is given in Figure 5 is an interesting case to document in detail. The program itself is not particularly difficult or involved, but it does illustrate the concepts of input, initialization, looping, computation, and output which are common to most programs.

The variables X_0 , and V_0 , and M define the initial position, initial velocity, and mass, respectively. D is the time increment used in the computation. N is the number of computations carried out between printout. L is the time limit on the computations.

The variable C is used as a counter to shift the program to the output statement every N times around the loop.

You should study the flow chart representation of the program until you are certain you understand the relation between the graph and the program. For introductory level programs it is generally not necessary to flow chart the problem first. If there is complicated branching, however, this is of great advantage.



Line Description

100]- States object of program.
110]- Sets variables X0, V0, M (the initial position, velocity and the mass)
120]- and variables D, N, L (D is time increment, N sets the number of passes
before printing, L is the time length).
130]- Sets variable F.
140]-
150]- Prints the heading.
160]-
170]- Sets the point counter, C, to the number of passes.
180]-
190]- Examines C: If equal to N then print T (time), V0, and X0 and reset
200]- C to zero.
210]-
220]- Computes new velocity and position.
230]-
240]- Sets V0 and X0 equal to V1 and X1.
250]-
260]- Increments C.
270]- Tests if $T \geq L$ and increments T by D or goes to end.
800]- Data for Variables.
801]-
999 End.

```
100  REM NEWTON'S SECOND LAW
110  READ X0,V0,M
120  READ D,N,L
130  LET F=2
140  PRINT
150  PRINT "T","V","X"
160  PRINT
170  LET C=N
180  FOR T=0 TO L STEP D
190  IF C<N THEN 220
200  PRINT T,V0,X0
210  LET C=0
220  LET V1=V0+F*D/M
230  LET X1=X0+V0*D
240  LET V0=V1
250  LET X0=X1
260  LET C=C+1
270  NEXT T
800  DATA 2,-4,1
801  DATA .1,10,5
999  END
```

RUN

T	V	X
0	-4	2
1	-2.	-1.1
2.	-1.84774E-06	-2.2
3.	2.	-1.3
4.	4.	1.59999

DONE

Scan Copyright ©
The Museum of HP Calculators
www.hpmuseum.org

Original content used with permission.

Thank you for supporting the Museum of HP
Calculators by purchasing this Scan!

Please do not make copies of this scan or
make it available on file sharing services.